
类 答案之书 : 窗口

	@布局配置([[根布局=真,宽度=-1,高度=-1,纵向布局=真,对齐方式=49,背景颜色=-1772801]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度=-1,高度=-1]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,对齐方式=17,@位于布局中间=真,字体大小=20,字体颜色=-11311497,粗体=真,内容="心中想着您的问题"]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=相对布局1,宽度=-1,高度DP=300,对齐方式=16,@位于布局中间=真]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,宽度DP=240,高度DP=300]])
	变量 图片框1 : 图片框
	@布局配置([[父布局=相对布局1,宽度=-1,高度=-1,图片资源="../../../素材/Png/答案之书_开始.png"]])
	变量 图片框3 : 图片框
	@布局配置([[父布局=相对布局1,图片资源="../../../素材/Png/答案之书_结束.png",@位于布局中间=真,横向偏移DP=-40,宽度DP=230,高度DP=330,可视=假]])
	变量 图片框2 : 图片框



	变量 时钟1:时钟
	变量 时钟2:时钟
	变量 时钟3:时钟
	变量 时钟4:时钟
	变量 时钟5:时钟
	变量 答案之书jSON:JSON数组
	变量 答案之书_内容:集合
	事件 答案之书:创建完毕()
		本对象.显示标题栏 = 假
		本对象.状态栏字体黑色 = 真
		本对象.导航栏颜色 = -1772801
		本对象.状态栏颜色=-1772801
		订阅事件()
		答案之书jSON=读入资源文件(本对象,"Divination/答案之书/答案之书.json")
		循环(索引,,答案之书jSON.长度)
			答案之书_内容.添加成员(答案之书jSON.取JSON对象(索引).取文本("内容"))
		结束 循环
	结束 事件

	变量 是否打开:逻辑型
	事件 线性布局1:被单击()
		如果 是否打开==真 则
			变量 消息 : DX基础对话框
			事件 消息:按钮1被单击() : 逻辑型
				图片框3.图片资源 = "../../../素材/Png/答案之书_开始.png"
				文本框1.可视 = 假
				图片框2.可视 = 假
				图片框3.可视 = 假
				序列帧=0
				图片框1.图片路径=""
				文本框1.内容=""
				时钟1.时钟周期 = 100
				图片框3.图片资源 = "../../../素材/Png/答案之书_结束.png"
				返回 假
			结束 事件
			事件 消息:按钮2被单击() : 逻辑型
				返回键被按下()
				返回 假
			结束 事件
			消息.样式(DX样式.IOS)
			消息.标题("您需要")
			消息.按钮1("合书重开")
			消息.按钮2("退出")
			消息.按钮3("取消")
			消息.按钮纵向显示(真)
			消息.可取消(假)
			消息.显示()
		否则
			图片框3.图片资源 = "../../../素材/Png/答案之书_开始.png"
			文本框1.可视 = 假
			图片框2.可视 = 假
			图片框3.可视 = 假
			序列帧=0
			图片框1.图片路径=""
			文本框1.内容=""
			时钟1.时钟周期 = 100
			图片框3.图片资源 = "../../../素材/Png/答案之书_结束.png"
			是否打开=真
		结束 如果

	结束 事件
	变量 音效延迟时钟:时钟
	事件 时钟1:周期事件()
		如果 图片框3.透明度 < 0.7 则
			图片框3.可视 = 假
			图片框1.可视 = 真
			图片框3.透明度 = 1
			时钟1.时钟周期 = 0
			时钟2.时钟周期 = 20
			音效延迟时钟.时钟周期 = 500
		否则
			图片框3.透明度=图片框3.透明度-0.1
		结束 如果
	结束 事件
	事件 音效延迟时钟:周期事件()
		音效延迟时钟.时钟周期 = 0
		媒体操作.播放assets音频(本对象,"mp3/答案之书_开始.mp3")
	结束 事件
	变量 序列帧:整数
	事件 时钟2:周期事件()
		如果 序列帧==80则
			时钟2.时钟周期 = 0
			时钟3.时钟周期 = 10
		否则
			序列帧=序列帧+1
			图片框1.位图对象 = 附加资源.取位图对象(本对象,"Divination/答案之书/答案之书/"+序列帧.到文本())
		结束 如果
	结束 事件

	事件 时钟3:周期事件()
		如果 图片框1.透明度 < 0.7 则
			图片框1.可视 = 假
			图片框1.透明度 = 1
			时钟3.时钟周期 = 0
			图片框2.可视=真
			时钟4.时钟周期 = 10

		否则
			图片框1.透明度=图片框1.透明度-0.1
		结束 如果
	结束 事件

	变量 宽度:整数
	变量 高度:整数
	变量 偏移:整数=-40

	事件 时钟4:周期事件()
		如果 图片框2.横向偏移 == 0 则
			媒体操作.播放assets音频(本对象,"mp3/答案之书_结束.mp3")
			时钟4.时钟周期 = 0
			图片框2.可视=假
			图片框3.可视 = 真
			文本框1.可视 = 真
			文本框1.内容=答案之书_内容[取随机数(0,答案之书jSON.长度)].到文本()
			//文本框1.内容="给蓝莓君打钱"
		否则
			宽度=宽度+6
			图片框2.宽度DP=宽度
			高度=高度+16
			图片框2.高度DP=高度
			偏移=偏移+2
			图片框2.横向偏移DP = 偏移
		结束 如果
	结束 事件

	@虚拟方法
	方法 返回键被按下()
		父对象.返回键被按下()
	结束 方法

结束 类